



HA-2600/02 SPICE OPERATIONAL AMPLIFIER MACRO-MODEL

Author: David Riemer

Introduction

This application note describes the SPICE macro-model for the HA-2600/02, wideband, high impedance op-amp. The model was designed to be compatible with the well known SPICE program developed by the University of California in hope that most simulation software vendors follow this basic format and syntax. A schematic of the macro-model, the Spice net listings and various simulated performance curves are included. The macro-model schematic includes node numbers to help relate the SPICE listing to the schematic. The model is designed to emulate a typical rather than a worst case part. Most AC and DC parameters are simulated. DC parameters for the HA-2602 macromodel are slightly degraded from those of the HA-2600, with the AC parameters remaining equivalent. Significant poles are included to give the most accurate AC and transient simulation with minimum complexity.

Model Description

Input Stage

DP and DN represent the differential input resistance. Input bias currents are created by I1 and offset current is modeled with FA. Source VN represents the input offset voltage. C1 limits slew rate. No input parasitics due to package capacitance and lead inductance are included.

Gain Stage

G2, R2, CC, GOL, and RD simulate open loop gain. CC is the macro-model dominant pole capacitor.

Poles

The two most significant poles of the HA-2600/02 are modeled by RC networks.

Output Stage

EX1, D1 and D2 model output current limiting. IH and IL are the power supply currents. DPH, DPL and GPS vary the supply currents based on the opamps output current. DL, DH, ECC and EEE provide voltage clamping on the output to simulate the typical output voltage swing. Some effects of output parasitics due to package capacitance and inductance are lumped with the poles.

Parameters Not Modeled

To maintain a simple macro-model not all op amp parameters are modeled. Most of the parameters not modeled are listed below:

- Temperature Effects
- Differential Voltage Restrictions.
- Input Voltage and Current Noise.
- Common Mode Restrictions
- Tolerances for Monte Carlo Analysis
- Power Supply Range

Harris Corporation, Semiconductor Sector, makes no warranty, either expressed or implied, including but not limited to any warranties or merchantability and fitness for a particular purpose using this macro-model. This macro-model is provided solely and exclusively for use by Harris customers to assist them in the assessment of this product for possible applications. In no event will Harris be liable for special, collateral, incidental, or consequential damages in connection with or arising out of the use of the macro-model. Although every effort has been made to ensure accuracy of the information contained in this application note, Harris assumes no responsibility for inadvertent errors. Harris reserves the right to make changes to the product and the macro-model without prior notice.

Spice Listing

```

*
*COPYRIGHT (C) 1991 HARRIS CORPORATION
* ALL RIGHTS RESERVED
*
*HA-2600 MACROMODEL
*REV: 8-01-91
*BY: D.W. RIEMER
*
*PINOUT      +IN  -IN  VCC  VEE  OUT
*
.SUBCKT HA2600 1 2 4 5 3
.MODEL DP D  IS=1E-14      N=+1.5417
.MODEL DN D  IS=+1.1428E-14N=+1.5417
.MODEL DV D  IS=+1.7684E-15N=.2
.MODEL D1 D  IS=1E-9       N=1
.MODEL D2 D  IS=1E-9       N=+1.0
.MODEL DX D  IS=1E-20      N=+30.0
*
*INPUT STAGE
*VALUE OF SOURCE VN MODELS VIO AND
MAY BE ADJUSTED AS DESIRED.
*
VP 1 6 0
VN 2 7 +5E-04
I1 8 0 +1.0714E-09
FA 2 0 VN +1.625
DP 6 8 DP
DN 7 8 DN
FP 9 0 VP +1.6427E+05
FN 0 9 VN +1.4373E+05
GC 0 9 8 0 +6.5568E-08
GPP 9 0 4 0 +6.5568E-08
GPN 9 0 5 0 +1.16599E-07
IRX 0 9 -7.3735E-07
RT 9 0 1.0
*
*POLES
*
EP1 10 0 9 0 1.0
RP1 10 11 +6.368E+01
CP1 11 0 1E-10
EP2 12 0 11 0 1.0
RP2 12 13 +4.5486E+01
CP2 13 0 1E-10
*
*OUTPUT STAGE
*
G2 0 14 13 0 1.0
R2 14 0 +6.5577E+02
CC 14 15 +2.2E-11
GOL 15 0 14 0 +1.2257E+03
RD 15 0 +90.0
DH 15 16 DV
DL 17 15 DV
ECC 16 0 POLY 1 4 0 -2.7 1.0
EEE 17 0 POLY 1 5 0 +2.9 1.0
IH 4 0 +3.0E-03
IL 0 5 +3.0E-03
GPS 18 0 15 3 +0.99999E-01
DPH 4 18 DX
DPL 18 5 DX
D1 15 19 D1
D2 19 15 D2
EX1 19 0 POLY 2 15 0 3 0 0.0 -8.8481E-01 +1.881
RO 15 3 +10.0
.ENDS HA2600

```


Spice Listing

```


*
*COPYRIGHT (C) 1991 HARRIS CORPORATION
*ALL RIGHTS RESERVED
*
*HA-2602 MACROMODEL
*REV: 8-01-91
*BY: D.W. RIEMER
*
*PINOUT      +IN  -IN  VCC  VEE  OUT
*
.SUBCKT HA2602 1 2 4 5 3
.MODEL DP D IS=1E-14 N=+1.1563
.MODEL DN D IS=+1.1428E-14N=+1.1563
.MODEL DV D IS=+1.061E-15 N=.2
.MODEL D1 D IS=1E-9 N=1
.MODEL D2 D IS=1E-9 N=+1.0
.MODEL DX D IS=1E-20 N=+30.0
*
*INPUT STAGE
*VALUE OF SOURCE VN MODELS VIO AND
MAY BE ADJUSTED AS DESIRED.
*
VP 1 6 0
VN 2 7 +2E-03
I1 8 0 +5.3571E-09
FA 2 0 VN +1.625
DP 6 8 DP
DN 7 8 DN
FP 9 0 VP +3.2853E+04
FN 0 9 VN +2.8747E+04
GC 0 9 8 0 +1.5546E-07
GPP 9 0 4 0 +4.9162E-08
GPN 9 0 5 0 +1.5546E-07
IRX 0 9 -1.5371E-06
RT 9 0 1.0
*
*POLES
*
EP1 10 0 9 0 1.0
RP1 10 11 +6.368E+01
CP1 11 0 1E-10
EP2 12 0 11 0 1.0
RP2 12 13 +4.5483E+01
CP2 13 0 1E-10
*
*OUTPUT STAGE
*
G2 0 14 13 0 1.0
R2 14 0 +6.5577E+02
CC 14 15 +2.2E-11
GOL 15 0 14 0 +7.3545E+02
RD 15 0 +90.0
DH 15 16 DV
DL 17 15 DV
ECC 16 0 POLY 1 4 0 -2.7 1.0
EEE 17 0 POLY 1 5 0 +2.9 1.0
IH 4 0 +3.5E-03
IL 0 5 +3.5E-03
GPS 18 0 15 3 +0.9999E-01
DPH 4 18 DX
DPL 18 5 DX
D1 15 19 D1
D2 19 15 D2
EX1 19 0 POLY 2 15 0 3 0 0.0 -8.4731E-01 +1.8436
RO 15 3 +10.0
.ENDS HA2602

```

Macro-Model Schematic




Model Performance



Model Performance (Continued)

SMALL SIGNAL RESPONSE



LARGE SIGNAL RESPONSE

